Fourier Synthesis of Triangle and Square Waves Passed Through Various Filters
Physics 238

Basic Waveforms

First, we will define three simple waves, each of amplitude Vin, and frequency f.
Use Fourier modes up to 'nmax'. We trade off computational efficiency for accuracy here. A value of '99' is reasonably quick but still gives good results.

In[1]:=
```
Clear["Global`*"]; nmax = 1000; DateString[]
```
Out[1]=
```
Wed 30 Apr 2025 11:01:03
```

Define several different wave forms. The input is given by a simple Fourier Series. The output is given by taking each term in the sum and passing it through a filter with gain a[f] and phase shift ϕ[f] (defined later).

Sinusoid

In[2]:=
```
sinusoid[Vin_, f_, t_] := Vin * Sin[2 π f t]
sineout[Vin_, f_, t_] := Vin *a[f] *Sin[2 π f t + ϕ[f]]
```

Triangle

In[4]:=
```
At[n_] := 8/π^2 1/n^2 (* Don't need to alternate signs if we use Cosine[] *)
```

In[5]:=
```
triangle[Vin_, f_, t_] := Vin * Sum[At[n] * Cos[n * 2 π f t], {n, 1, nmax, 2}];
```

In[6]:=
```
triangleout[Vin_, f_, t_] :=
  Vin * Sum[At[n] *a[n*f] *Cos[n * 2 π f t + ϕ[n*f]], {n, 1, nmax, 2}];
```

Square

In[7]:=
```
Asq[n_] := 4/π 1/n
```

In[8]:=
```
square[Vin_, f_, t_] := Vin Sum[Asq[n] * Sin[n 2 π f t], {n, 1, nmax, 2}]
squareout[Vin_, f_, t_] :=
 Vin Sum[Asq[n] ×a[n f] Sin[n 2 π f t + ϕ[n f]], {n, 1, nmax, 2}]
```

Sawtooth

In[10]:=
$$\text{Asaw}[n\_] := \frac{-1}{n\,\pi}$$

In[11]:=
```
saw[Vin_, f_, t_] := Vin Sum[Asaw[n] * Sin[n 2 π f t], {n, 1, nmax, 1}]
sawout[Vin_, f_, t_] :=
 Vin Sum[Asaw[n] × a[n f] * Sin[n 2 π f t + ϕ[n f]], {n, 1, nmax, 1}]
```

Component Values

It's easiest to scale everything by the breakpoint frequency, fb. For our typical experiment with R = 3.3 kΩ, C=22 nF, fb = 2192 Hz.

In[13]:=
```
ωb = 1 / (3300 * 22. * 10^(-9));
fb = ωb / (2 π)
```

Out[14]=
```
2192.22
```

Utility Functions

We will also use a simple helper function for the period T.

In[15]:=
```
T[f_] := 1 / f
```

This is a utility function to compare two wave forms.

In[82]:=
```
compare[vin_, vout_, f_] := Plot[{vin[10, f, t], vout[10, f, t]},
  {t, 0, 2 T[f]}, PlotLegends → {"vin", "vout"}, PlotRange → All,
  PlotLabel → StringForm["f = `` fb", f / fb], ImageSize → 300,
  Frame → False, AxesLabel → {"t (s)", "V (Volts)"}, LabelStyle → Larger]
```

Show a column of 3 graphs, tightly spaced. Tweak/experiment with the ImageSize, Spacings, and AspectRatio to get it all to work.

In[83]:=
```
showGraphs[list_] := GraphicsColumn[list,
  Alignment → {Center, Center}, Frame → All,
  Spacings → 0, ImageSize → 400, AspectRatio → 3 / GoldenRatio]
```

In[84]:=
```
showImg[file_] := Show[Import[file], ImageSize → 300]
```

Low-pass Filter

In[85]:=
```
a[f_] := 1 / Sqrt[1 + (f / fb)^2]
ϕ[f_] := ArcTan[-f / fb]
```

Sinusoid

In[87]:=
```
showGraphs[Table[compare[sinusoid, sineout, f], {f, {fb / 10, fb, 10 fb}}]]
```

Out[87]=



Triangle

In[88]:=
```
showGraphs[Table[compare[triangle, triangleout, f], {f, {fb / 10, fb, 10 fb}}]]
```

Out[88]=



In[58]:=
```
SetDirectory[NotebookDirectory[]];
```

In[92]:=
```
showGraphs[{
  Import["Triangle-LP-Low-f.png"],
  Import["Triangle-LP-Med-f.png"], Import["Triangle-LP-High-f.png"]}]
```

Out[92]=



TBS1052C - 11:11:59 AM 4/30/2025



TBS1052C - 11:15:07 AM 4/30/2025



TBS1052C - 11:15:48 AM 4/30/2025

Square

In[24]:=
```
showGraphs[Table[compare[square, squareout, f], {f, {fb / 10, fb, 10 fb}}]]
```

Out[24]=



In[93]:=
```
SetDirectory[NotebookDirectory[]];
showGraphs[{
  Import["Square-LP-Low-f.png"],
  Import["Square-LP-Med-f.png"], Import["Square-LP-High-f.png"]}]
```

Out[94]=



TBS1052C - 4:01:38 PM 4/29/2025



TBS1052C - 10:56:04 AM 4/30/2025



TBS1052C - 3:53:24 PM 4/29/2025

Sawtooth

In[26]:=
```
showGraphs[Table[compare[saw, sawout, f], {f, {fb / 10, fb, 10 fb}}]]
```

Out[26]=



High-pass Filter

In[27]:=
```
a[f_] := (f / fb) / Sqrt[1 + (f / fb) ^ 2]
ϕ[f_] := ArcTan[f, fb]
```

Sinusoid

In[29]:=
```
showGraphs[Table[compare[sinusoid, sineout, f], {f, {fb / 10, fb, 10 fb}}]]
```

Out[29]=



Triangle

In[30]:=
```
showGraphs[Table[compare[triangle, triangleout, f], {f, {fb / 10, fb, 10 fb}}]]
```

Out[30]=



Square

In[31]:=
```
showGraphs[Table[compare[square, squareout, f], {f, {fb / 10, fb, 10 fb}}]]
```

Out[31]=



Sawtooth

In[32]:=
```
showGraphs[Table[compare[saw, sawout, f], {f, {fb / 10, fb, 10 fb}}]]
```

Out[32]=